Snake Game part 1

Processing Practice Project

Let's make a simple game to practice what we've learned so far.

Game Concept

The concept of this game is a simple remake of the Snake Game, which was one of the first games to appear on a cell phone.

- The player controls a "snake" on a grid.
- The snake begins as a square one grid block in size.
- Red "food" blocks appear randomly on the screen.
- Each time the snake eats a food block, it gets bigger, gaining another block on its tail.
- If the snake crashes into a wall, or into its own tail, the game is over.
- The goal is to grow as large as possible without crashing.

Step by Step

The most important thing when doing a project like this, is to break it down into small, simple steps. Then, write code for each step.

Let's begin by writing a program that displays a green square, on a black background. This will be the beginning our snake "head".

Step 1: Snake Head

```
void setup() {
  size(600, 600);
}

void draw() {
  background(0);
  fill(100, 215, 0);
  square(100, 200, 30);
}
```

Setup Grid with Variables

Great! We have a snake head. Now, we need to do a little planning. We don't want the snake to move smoothly across the board. Instead, we want a sort of grid, where the snake moves one square at a time. It cannot appear in between squares.

Let's setup a variable to define our grid size, and use these to position the snake on the grid.

```
int gridSize = 20;
```

We will also use a new variable type, called **PVector**. This is an object that holds both and x and a y value, so we can keep the location of our snake in one variable, instead of having separate x, and y variables. This will also make it easier later when we add the tail.

```
PVector snake;
```

Let's create a method called `newGame()` to set the starting location of our snake. Later, we can use this method to reset all of the variables we need for a new game. While we're at it, lets move the job of drawing the snake to its own method, `showSnake()`.

```
void newGame() {
  snake = new PVector(width/2, height/2);
  snakeSize = gridSize;
}
void showSnake() {
  fill(100, 215, 0);
  square(snake.x, snake.y, snakeSize);
}
```

The code should now look like this:

```
int gridSize = 20;

PVector snake;
int snakeSize;

void setup() {
  size(600, 600);
  newGame();
}

void draw() {
  background(0);
  showSnake();
}

void newGame() {
  snake = new PVector(width/2, height/2);
  snakeSize = gridSize;
}

void showSnake() {
  fill(100, 215, 0);
  square(snake.x, snake.y, snakeSize);
}
```

Challenges

1. Change the snake and food to be circles, instead of squares.